EXAMPLE 2: consumer optimization problem

We have agent, who has
E: like Endowement
    X Quantity = 1
    Y Quantity = 1
and spend his income on
D: like Demand
    X Quantity = 1
    Y Quantity = 1

Precision of MPSGE is 6 decimals by default. If we want to use other precision, we have to redefine it via OPTION.

Utility function is the Cobb-Douglas function $U(x,y)=X^{1/4}*Y$ or $U(x,y)=X*Y^4$ with MRS=(1/4)/1
No production implies that X=const and Y=const => Prices are determined just by consumer demand.
MRS(1,1) = y/4x = 1/4 = Px/Py describes endowment => **X=Y** and **4Px=Py**

The budget constraint is Px*1+ Py*1 = Px*X + Py*Y
Px*1+4Px*1 = Px*X +4Px*X
5Px =5Px*X
X = 1 => Y = 1

income/PX=5
income/Py=1.25




```
SCALAR
        X       QUANTITY OF X FOR WHICH THE MRS IS TO BE EVALUATED /1/
        Y       QUANTITY OF Y FOR WHICH THE MRS IS TO BE EVALUATED /1/
        MRS     COMPUTED MARGINAL RATE OF SUBSTITUTION;
$ONTEXT
$MODEL:MRSCAL

$COMMODITIES:
        PX      ! PRICE INDEX FOR GOOD X
        PY      ! PRICE INDEX FOR GOOD Y

$CONSUMERS:
        RA      ! REPRESENTATIVE AGENT INCOME

$DEMAND:RA      s:1
        D:PX    Q:1     P:(1/4)
        D:PY    Q:1     P:1
        E:PX    Q:X
        E:PY    Q:Y

$OFFTEXT
$SYSINCLUDE mpsgeset MRSCAL

$INCLUDE MRSCAL.GEN
SOLVE MRSCAL USING MCP;

* Following the solution, we compute a function of the solution values
*(the ratio of the PX to the PY and storing this result in the scalar MRS).
MRS = PX.L / PY.L;
OPTION MRS:8;
DISPLAY MRS;
```

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR PX	.	0.400	+INF	.
---- VAR PY	.	1.600	+INF	.
---- VAR RA	.	2.000	+INF	.

Conclusion: consumers income is the total endowment. The final prices should reflect the calibration point and MRS, when no producers.

Supplement Material to EXAMPLE 2:

1) Compare alternative ways to normalize the model:

Case 1: PX.L=2 and PY.L=1 **Case 3:** PX.L=2 and PY.L=2
Case 2: PX.L=1 and PY.L=2 **Case 4:** RA.L=1

SOLUTION

Normalization is what the model assumes as a starting number. The original example assumes that all variables (except income) are normalized to 1 (this is the default setting in MPSGE).

MRS(1,1)=1/4 => MRS=y/4x, i.e. consumer prefers Y over X => PY=4PX
MRS(1,1) => y=x, i.e. consumer possesses Y=X
no trade => demand=endowment

Income balance: PX*1 + PY*1 = PX*1 + PY*1
Expenditures divided by PY: PX/PY*1 + PY/PY*1 = ¼ *1+ 1*1 = 5/4 => Expenditures=5/4 * PY

Case original

PX.L=1
PY.L=1

Income: PX*1 + PY*1 = 1*1 + 1*1 =2
Income balance: 5/4* PY = 2

=> PY=2*4/5 = 1.6
PX= ¼ *PY=2/5=0.4



Case 1

PX.L=2
PY.L=1

Income: PX*1 + PY*1 = 2*1 + 1*1 =3
Income balance: 5/4* PY = 3

=> PY=3*4/5 = 2.4
PX= ¼* PY=3/5=0.6



Case 2

PX.L=1
PY.L=2

Income: PX*1+PY*1= 1*1+2*1 = 3
Income balance: 5/4*PY=3

=> **the same results as in Case 1**

part of the MPSGE code:

Case original

```
$SYSINCLUDE mpsgeset DEMAND

$INCLUDE DEMAND.GEN
SOLVE DEMAND USING MCP;
```

RESULT:

```
---- VAR PX          .          0.400
---- VAR PY          .          1.600
---- VAR RA          .          2.000
```

Case 1

```
$SYSINCLUDE mpsgeset MRSCAL
PX.L=2;
$INCLUDE MRSCAL.GEN
SOLVE MRSCAL USING MCP;
```

RESULT:

```
---- VAR PX          .          0.600
---- VAR PY          .          2.400
---- VAR RA          .          3.000
```

Case 2

```
$SYSINCLUDE mpsgeset MRSCAL
PY.L=2;
$INCLUDE MRSCAL.GEN
SOLVE MRSCAL USING MC
```

RESULT:

```
---- VAR PX     .     0.600
---- VAR PY     .     2.400
---- VAR RA     .     3.000
```

Case 3

PX.L=2
PY.L=2

Income: PX*1+PY*1=2*1+2*1=4
Income balance: 5/4*PY=4

=> PY=4*4/5=3.2
PX= ¼ *PY= 0.8



Case 4

RA.L=1

Income: PX*1+PY*1= 1
Income balance: 5/4*PY=1

=> PY= 1/(5/4)= 0.8
PX= ¼ *PY= 0.2



part of the MPSGE code:

Case 3

```
$SYSINCLUDE mpsgeset MRSCAL
PX.L=2;
PY.L=2;
$INCLUDE MRSCAL.GEN
SOLVE MRSCAL USING MCP;
```

RESULT:

```
---- VAR PX          .          0.800
---- VAR PY          .          3.200
---- VAR RA          .          4.000
```

Case 4

```
$SYSINCLUDE mpsgeset MRSCAL
RA.FX=1;
$INCLUDE MRSCAL.GEN
SOLVE MRSCAL USING MCP;
```

RESULT:

```
---- VAR PX          .          0.200
---- VAR PY          .          0.800
---- VAR RA        1.000        1.000        1.000
```

Since numeraire is RA (by default setting in MPSGE), normalization of this variable requires to fix it (i.e. RA.FX instead of RA.L)

Conclusion: **(i)** Normalization of variables does not change variables in real terms, but only in nominal terms (similar to numeraire). Normalization only helps solver to find a solution. **(ii)** Different normalisations may generate the same results in nominal variables if their impact on values of variables is the same (like in Cases 1 and 2). **(iii)** Changing normalization of variables that represent values (like income) requires fixing that variable.

2) Modify the model to obtain Px=Py

SOLUTION

Consumer has the same amount of goods (calibration point) in the original model, but he prefers 4 times more Y than X

⇓

Demand on Y goes up

⇓

Py goes up

Original model

- MRS(1,1) = ¼ => consumer prefers Y over X if Px=Py
- no producers => Px/Py=¼
- MRS = Y/(4X) => X=Y because Px<Py

⇓

X=Y and 4Px=Py

max $U = XY^4$ **s.t. 0.4*X+1.6*Y=2**

⇓

Px*X+4Px*X=2 => Px*X=0.4 => X=1=Y

- $MRS_{1,1}$ = ¼ => consumer **initially** prefers 4 times more Y than X, but he possess equal amount of goods

⇓

$$\frac{X}{Y} = \frac{1}{1}$$

⇓

X=Y and 4Px = Py

⇓

final allocation = initial allocation



Case 1

- Px=Py => Px/Py = 1 = MRS(1,1)
- Px=Py= 1 => RA=Px*1+Py*1 = 2
- MRS(1,1)= 1 => consumer prefers the same quantity of Y and X based on his utility function => U=XY

⇓

X=Y and Px=Py

max $U = XY$ **s.t. X+Y=2**

⇓

Px*X+Px*X=2 => Px*X=1 => X=1=Y

- $MRS_{x,y}$ = ¼ => consumer **initially** prefers 4 times more Y than X based on his calibration point

⇓

his initial allocation must changes

MRS=X/Y *¼=1

$MRS_{4,1}$= ¼ = Px/Py => $\frac{X}{Y} = \frac{4}{1}$

⇓

X=4Y and 4Px = Py

⇓

Px*X+4Px*¼X=2 => Px*X=1 => X=1, Y=¼



Case 2

- Px=Py => Px/Py = 1 = MRS(1,1)
- Px=Py= 1 => RA=Px*1+Py*1 = 2
- MRS(1,1)= 1 => consumer prefers the same quantity of Y and X based on his utility function => U=XY

⇓

X=Y and Px=Py

max $U = XY$ **s.t. X+Y=2**

⇓

Px*X+Px*X=2 => Px*X=1 => X=1=Y

- $MRS_{x,y}$ = 1 => consumer **initially** prefers the same quantity of Y and X and he possess equal amount of goods

⇓

final allocation = initial allocation

⇓

X=Y and Px = Py

⇓

Px*X+Px*X=2 => Px*X=1 => X=1, Y=1



part of the MPSGE code:

<table>
<tr>
<td>

```
$DEMAND:RA   s:1
D:PX    Q:1     P:(1/4)
D:PY    Q:1     P:1
E:PX    Q:1
E:PY    Q:1
```

RESULT:

```
VAR PX     0.400
VAR PY     1.600
VAR RA     2.000
```

</td>
<td>

```
$DEMAND:RA   s:1
D:PX    Q:4     P:(1/4)
D:PY    Q:1     P:1
E:PX    Q:1
E:PY    Q:1
```

RESULT:

```
VAR PX     1.000
VAR PY     1.000
VAR RA     2.000
```

</td>
<td>

```
$DEMAND:RA  s:1
D:PX    Q:1     P:1
D:PY    Q:1     P:1
E:PX    Q:1
E:PY    Q:1
```

RESULT:

```
VAR PX     1.000
VAR PY     1.000
VAR RA     2.000
```

</td>
</tr>
</table>

Conclusion: Equal prices can be obtained either through modification of **(i)** calibration point (Case 1) or **(ii)** MRS (Case 2). Both changes means modification of utility function.

Exercise 2A:

Show that the demand function is homothetic by uniform scaling of the
x and y endowments (for instance 2). The resulting MRS should remain unchanged: MRS(2,2) = y/4x = 1/4 = PX/PY describes endowment, while MRS(1,1) describes preferred allocation.

The new budget constraint is 2+2 = PX*2*1 + PY*2*1, since income rescaling requires to rescale both side of the budget equation. Thus 4/PY = PX/PY * 2+ PY/PY * 2 =10/4
This gives us PY=8/5 =1.6




```
SCALAR
        X       QUANTITY OF X FOR WHICH THE MRS IS TO BE EVALUATED /2/
        Y       QUANTITY OF Y FOR WHICH THE MRS IS TO BE EVALUATED /2/
        MRS     COMPUTED MARGINAL RATE OF SUBSTITUTION;

$ONTEXT

$MODEL:MRSCAL

$COMMODITIES:
        PX      ! PRICE INDEX FOR GOOD X
        PY      ! PRICE INDEX FOR GOOD Y

$CONSUMERS:
        RA      ! REPRESENTATIVE AGENT INCOME

$DEMAND:RA      s:1
        D:PX    Q:1     P:(1/4)
        D:PY    Q:1     P:1
        E:PX    Q:X
        E:PY    Q:Y

$OFFTEXT
$SYSINCLUDE mpsgeset MRSCAL

$INCLUDE MRSCAL.GEN
SOLVE MRSCAL USING MCP;
```

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR PX	.	0.400	+INF	.
---- VAR PY	.	1.600	+INF	.
---- VAR RA	.	4.000	+INF	.

Conclusion: this is homothetic function, because only total income has changed (prices = const) after rescaling of endowment

Exercise 2B:

Modify the demand function calibration point so that the reference prices of both x and y equal unity.

The relationship of initial prices (PX=1/4 and PY=1) is equal now to MRS(4,1)=1/4 instead of MRS(1,1)=1/4.

MRS=1/4 means that consumer prefers Y four times more than X. However, his preferred allocation is to buy X four times more than Y (while supply of both goods is the same). It gives 4*(1/4)=1, i.e. consumer finally prefers X=Y and his utility function becomes U(x,y)=x*y.

MRS(1,1) = y/x = 1/1 =PX/PY, i.e. PX=PY
We put this into budget constraint 2 = PX*1 + PY*1 = PX+PY = 2PX => **PX = 1** => **PY = 1**

MRS(4,1) = y/x = 1/4. This means that **X=4Y**. We put this into budget constraint:
2 = PX*X + PY*Y = 1*4y + 1*y = 5y
In order to get PX=PY=1 (budget line crosses horizontal and vertical lines in 2=2/1), we have to use y=2/5=0.4 and x=4*0.4=1.6.

Thus if **MRS(4,1)**=1/4 & MRS(1,1)=1=PX/PY => U(x,y)=x*y
MRS(1,1)=1/4=PX/PY => MRS(4,1)=1/16 => U(x,y)=$x^{1/4}$*y or U(x,y)=x*y^4




```
SCALAR
        X       QUANTITY OF X FOR WHICH THE MRS IS TO BE EVALUATED /1/
        Y       QUANTITY OF Y FOR WHICH THE MRS IS TO BE EVALUATED /1/
        MRS     COMPUTED MARGINAL RATE OF SUBSTITUTION;

$ONTEXT
$MODEL:MRSCAL
$COMMODITIES:
        PX      ! PRICE INDEX FOR GOOD X
        PY      ! PRICE INDEX FOR GOOD Y
$CONSUMERS:
        RA      ! REPRESENTATIVE AGENT INCOME

$DEMAND:RA      s:1
        D:PX    Q:4     P:(1/4)
        D:PY    Q:1     P:1
        E:PX    Q:X
        E:PY    Q:Y
$OFFTEXT
$SYSINCLUDE mpsgeset MRSCAL
$INCLUDE MRSCAL.GEN
SOLVE MRSCAL USING MCP;
```

	LOWER	LEVEL	UPPER	MARGINAL
---- VAR PX	.	1.000	+INF	.
---- VAR PY	.	1.000	+INF	.
---- VAR RA	.	2.000	+INF	.

Conclusion: Relationship between prices depends on supply and MRS, but not on calibration point